Stevin : 25.6.89.14/A2 COMPUTERPROGRAMMA VOOR HET BEREKENEN VAN SPANNINGEN IN EEN UIT MEERLAGEN OPGEBOUWDE BALK

mei 1989 M.H.Kolstein, D.Wiglema

KEYWORDS: Rekenmodel, meerlagen-balk

TU-Delft
Delft University of Technology
Faculty of Civil Engineering
Stevin laboratory - Steel constructions
P.O. Box 5049 ; 2600 GA DELFT
tel. 015-783729 / 015-784005

INHOUDSOPGAVE

1. INLEIDING. 1

2. REKENTECHNISCHE GRONDSLAG. 2
2.1. Randvoorwaarden. 5
2.1.1. Schuifspanningen. 5
2.1.2. Rekken. 6
2.1.3. Hellingshoeken. 6
2.2. Uitwerking berekening. 7

3. COMPUTERPROGRAMMA MULTIC 10

4. VERGELIJKING MULTIC-SMITH. 11
4.1. Invoerdata vergelijking TesTT-Smith. 12
4.2. Vergelijking TesTT-Multic. 17

5. INVLOED LENGTE-INVOER OP BEREKENINGSRESULTATEN. 18

6. WERKING/MOGELIJKHEDEN. 21
6.1. Algemeen. 21
6.2. Rekenvoorbeelden. 21

7. REFERENTIES 26

LIJST VAN TABELLEN

LIJST VAN FIGUREN

Figuur 1. Tweelagige balkconstructie. 2
Figuur 2. Meerlagige balkconstructie. 3
Figuur 3. Spanningsverdeling per balk. 3
Figuur 4. Spanningsverloop prismatische doorsneden. 4
Figuur 5. Hoekverdraaiing contactvlak. 7
Figuur 6. Vergelijking MULTIC-SMITH. 11
Figuur 7-1. Vergelijking TesTT-Smith. 14
Figuur 8. Balkopbouw TesTT en Multic 17
Figuur 9. Verband Ki en E/d-hechtlaag. 18
Figuur 10. Lengte-invloed op rekkenverloop. 19
Figuur 11. Vergelijk 4-punts- en 3-puntsbuigproef. 20

BIJLAGEN

BIJLAGE A. Reken-matrix "Multic" (3-laags-constructie). 27
BIJLAGE B. Programma MULTIC 28
BIJLAGE C. Hulpprogramma TesTT 38
BIJLAGE D. Rekenvoorbeeld HE 100 B 39
BIJLAGE E. Rekenvoorbeeld ZOAB-slijtlaag 40

1. INLEIDING.

Het in dit rapport behandelde rekenmodel is gemaakt voor de berekening van spanningen en rekken in meerlagige asfaltslijtlagen op orthotrope stalen plaatbruggen. De ontwikkeling en toepassing van deze orthotrope stalen plaatbruggen is de laatste 25 jaar enorm toegenommen. Op het gebied van brugdekbekledingen brengt deze constructievorm echter nieuwe problemen met zich mee. Door de grote flexibiliteit van dit type brugdek moeten namelijk hoge eisen worden gesteld aan het brugdekbekledingsmeteriaal. Gietasfaltconstructies met een dikte van 50 mm worden op dit moment het meest toegepast. Er treedt echter vaak scheurvorming op ter plaatse van de verstijvingen in het stalen dek. Deze scheurvorming wordt aangemerkt als een ernstig schadeverschijnsel,enerzijds omdat het ontstaan van scheuren een directe bedreiging voor de onderliggende draagconstructie betekent (roestvorming e.d.), anderzijds omdat scheuren vaak andere schadeverschijnselen inleiden, zoals vervormingen, afrukking, opstijging e.d. De traditionele slijtlaag-constructie, opgebouwd uit een top- en een onderlaag van gietasfalt, een lak-, kleef-, en mastieklaag, vertoont veelal scheurvorming die zich in al deze lagen tot op het stalen dek doorzet. Om deze problematiek te kunnen bestuderen en te zoeken naar een verbeterde slijtlaag is door de Faculteit der Civiele Techniek een onderzoeksvoorstel ingediend bij de Stichting voor Technische Wetenschappen, getiteld: "Optimalisatie van economische bruikbaarheid van dikke slijt- en isolatielagen op stalen brugdekken (criteria en factoren) [1].

In [2,3,4] is bij het onderzoek de nadruk gelegd op de optimalisatie van de gietasfalt-deklaag. Hiertoe zijn een aantal gemodificeerde gietasfaltmengsels gekozen en beproefd waarbij de resultaten vergeleken zijn met die van de traditionele door Rijkswaterstaat voorgeschreven gietasfaltmengsels.

Naast de beproevingen zijn ook rek-berekeningen gemaakt waarbij gekeken is naar de relatie tussen de stijfheidsmoduli van het asfalt en de optreden rekken in de constructie. Dit kan van groot belang zijn voor het voorspellen van optredende rekken in een asfaltconstructie. Er wordt vooralsnog van uitgegaan dat de vermoeiingssterkte afhankelijk is van de optredende rekken.

Omdat we bij dergelijke asfalt/staal-constructies te maken hebben met 2 verschillende elasticiteismoduli is het alleen mogelijk simpele constructies snel handmatig door te rekenen waardoor de werkelijkheid slecht benaderd wordt. Bij het model "Smith" [6] waarvan in een eerder stadium van het onderzoek gebruik is gemaakt kan naast de 2 verschillende constructielagen ook een variabele schuifstijfheidscoefficient (Ki) worden ingevoerd. Deze coefficient geeft de mate van samenwerking aan tussen de 2 constructielagen.

2. REKENTECHNISCHE GRONDSLAG.

Bij het rekenmodel "Smith" wordt uitgegaan van een 2-lagige balkconstructie volgens fig.1. De verbindingslaag tussen de 2 lagen is van grote invloed op het optredende spanningsverloop in een doorsnede bij belasting. Het spreekt vanzelf dat bij het berekenen van spanningen en rekken in een slijtlaagopbouw uit de praktijk deze constructie sterk geschematiseerd moet worden. Zeker bij constructielagen met sterk varierende elasticiteitsmoduli is het moeilijk een acceptabele schematisereing te vinden. Ook is het moeilijk de praktische waarde voor de schuifstijheidscoefficient Ki te bepalen.



Figuur 1. Tweelagige balkconstructie.

Bij het nieuwe model is het aantal lagen uitgebreid tot 20 (afhankelijk van de gebruikte rekenapparatuur). Voor de bespreking van de opbouw van het model wordt uitgegaan van een 3-laags constructie.

De berekening van de asfalt/staalbalk wordt vlgs. de elasticiteitstheorie uitgevoerd. Hierbij wordt er van uitgegaan dat bij constante temperatuur, belasting en belastingstijd bitumineuze materialen zich elastisch gedragen.
Bij het berekenen van spanningen in een doorsnede van de meerlagige balk wordt uitgegaan van een eenzijdig ingeklemde ligger belast met een puntlast (P) op het vrije uiteinde (fig.2).



Figuur 2. Meerlagige balkconstructie.

De meerlagige balk wordt beschouwd als een aantal afzonderlijke balkjes waarbij voor de contactvlakken per balkje verschillende randvoorwaarden gelden m.b.t. tot optredende spanningen en verplaatsingen. Voor het bepalen van de spanningsverdelingen wordt eerst de spanningsverdeling in een afzonderlijk balkje beschouwd (fig.3).



Figuur 3. Spanningsverdeling per balk.

Spanningen in de z-richting worden verwaarloosd.
Het 2-dimensionale spanningsbeeld in balken met prismatische doorsneden kan verkregen worden door integratie van de volgende spanningsvergelijking [5].

$$\frac{\delta^4\phi}{\delta x^4} + \frac{2\delta^4\phi}{\delta x^2 \delta y^2} + \frac{\delta^4\phi}{\delta y^4} = 0$$

waarin ϕ een spanning-functie is zodat :

$$\sigma_x = \frac{\delta^2\phi}{\delta y^2} \; ; \quad \sigma_y = \frac{\delta^2\phi}{\delta x^2} \; ; \quad \tau_{xy} = \frac{\delta^2\phi}{\delta x dy}$$

De volgende spanningsfuntie geeft aannemelijke resultaten voor de eenzijdig ingeklemde ligger:

$$\phi = \frac{\alpha}{6} xy^3 \; + \; \beta\, xy \; + \; \frac{\gamma}{2} xy^2 \qquad \text{zodat:}$$

$$\sigma_x = \alpha xy + \gamma x \quad ; \quad \sigma_y = 0 \quad ; \quad \tau_{xy} = -\frac{\alpha}{2} y^2 - \beta - \gamma y$$

Het spanningsverloop van σ_x verloopt lineair over de hoogte van de beschouwde doorsnede (fig.4a). De schuifspanning verloopt volgens fig.4b.



(4a)



(4b)

Figuur 4. Spanningsverloop prismatische doorsneden.

Het voorafgaande geldt voor een enkele balk, voor een meerlagensysteem geldt voor elke laag de voorafgaande spanningsfuntie ϕ met voor de n^e laag de constanten :

$$\alpha_n \;,\; \beta_n \;,\; \gamma_n$$

Voor een willekeurige laag geldt dus:

$$\sigma_{x,n} = \alpha_n x y_n + \gamma_n y_n$$

$$\sigma_{y,n} = 0$$

$$\tau_{xy,n} = -\frac{\alpha_n}{2} y_n^2 - \beta_n - \gamma_n y_n$$

Door nu voor de verschillende lagen de randvoorwaarden voor optredende spanningen en verplaatsingen cq. rekken vast te leggen vinden we een systeem van (3*N) vergelijkingen met de onbekenden $\alpha_{1..N}$,$\beta_{1..N}$ en $\gamma_{1..N}$, waarbij n het aantal lagen is.
Hieruit kunnen de constanten bepaald worden en de spanningen berekend.

2.1. Randvoorwaarden.

Bij het bepalen van de randvoorwaarden is de hoofdgedachte dat de constructielagen niet verschuiven t.o.v. elkaar. Dit houdt in dat de rekken en schuifspanningen in de grensvlakken niet discontinu verlopen over de hoogte van de balk.
De volgende randvoorwaarden kunnen worden opgesteld:

2.1.1. Schuifspanningen.

De boven en onderrand van de meerlagenbalk zijn vrij van schuifspanningen. Dus:

$$\tau_1 \;(y_1 = -h_1) = 0 \qquad \ldots\ldots\ldots\ldots\ldots\ldots \quad (1)$$

$$\tau_2 \;(y_3 = h_3) = 0 \qquad \ldots\ldots\ldots\ldots\ldots\ldots \quad (2)$$

Schuifspanningen zijn in 2 aangrenzende lagen ter plaatse van het contactvlak gelijk aan elkaar.

$$\tau_1\ (y_1 = h_1) = \tau_2\ (y_2 = -h_2) \qquad \ldots\ldots\ldots\ (3)$$

$$\tau_2\ (y_2 = h_2) = \tau_3\ (y_3 = -h_3) \qquad \ldots\ldots\ldots\ (4)$$

De sommatie van alle schuifkrachten over de hoogte van de totale ligger in een willekeurige dwarsdoorsnede is gelijk aan de uitwendige belasting P.

$$\Sigma\tau_{xy}dy = P$$

$$\int \tau_1 + \int \tau_2 + \int \tau_3 = P$$

2.1.2. Rekken.

De rekken in 2 aangrenzende lagen zijn ter plaatse van het contactvlak gekijk aan elkaar. Er is vanuitgegaan dat er geen verplaatsing van lagen ten opzichte van elkaar is opgetreden.

$$\epsilon_{x_1}\ (y_1 = h_1) = \epsilon_{x_2}\ (y_2 = -h_2) \qquad \ldots\ldots\ldots\ (6)$$

$$\epsilon_{x_2}\ (y_2 = h_2) = \epsilon_{x_3}\ (y_3 = -h_3) \qquad \ldots\ldots\ldots\ (7)$$

Voor de rek in x-richting in een willekeurige dwarsdoorsnede van de balk geldt:

$$\epsilon_x = \frac{\sigma_x}{E} = \frac{\alpha xy}{E} + \frac{\gamma x}{E}$$

2.1.3. Hellingshoeken.

Op de grensvlakken van 2 lagen zijn de hellingshoeken van 2 lagen t.g.v. doorbuiging van de totale balk gelijk aan elkaar. Hierdoor zijn ook de hoekverdraaiingen van de liggers t.p.v. het contactvlak gelijk aan elkaar.

Voor verplaatsingen in y-richting geldt (vlgs [6]):

$$V = -\frac{\nu}{E}\left(\frac{\alpha}{2}xy^2 + \gamma xy\right) - \frac{\alpha}{6E}x^3 + \frac{\alpha}{2E}L^2 + \frac{2\beta(1+\nu)(L-x)}{E} - \frac{\alpha}{3E}L^2$$

Voor de hoekverdraaiing geldt vlgs. fig.5 :

$$\frac{dV}{dx} = -\frac{\nu}{E}(\frac{\alpha}{2}y^2 + \gamma y\ ) - \frac{\alpha x^2}{2E} + \frac{\alpha L^2}{2E} + \frac{2\beta(1+\nu)}{E}$$



Figuur 5. Hoekverdraaiing contactvlak.

Voor de hellingshoeken in de contactvlakken geldt:

$$\frac{dV_1}{dx}\,(y_1 = h_1) = \frac{dV_2}{dx}\,(y_2 = -h_2) \quad \ldots\ldots(8)$$

$$\frac{dV_2}{dx}\,(y_2 = h_2) = \frac{dV_3}{dx}\,(y_3 = -h_3) \quad \ldots\ldots(9)$$

Voor iedere randvoorwaarde kan een vergelijking worden opgesteld. Uit het geheel van de (in dit geval) 9 vergelijkingen met de 9 onbekenden kunnen de onbekende constanten α,β en γ d.m.v. matrix-berekening berekend worden.

2.2. Uitwerking berekening.

Aan de hand van de bepaalde randvoorwaarden kunnen de 9 vergelijkingen bepaald worden. De nummering correspondeerd met die uit hfdst.2.1.

1)

$$\tau_1\ (y_1 = -h_1) = 0 \quad \rightarrow$$

$$-\frac{\alpha_1}{2}\,h_1^2 - \beta_1 + \gamma_1 h_1 = 0 \quad \rightarrow$$

$$-\frac{h_1^2}{2}\,\alpha_1 - \beta_1 + h_1\gamma_1 = 0$$

2)

$$\tau_3\ (y_3 = h_3) = 0 \quad \rightarrow$$

$$-\frac{\alpha_3}{2}\,h_3^2 - \beta_3 - \gamma_3 h_3 = 0 \quad \rightarrow$$

$$-\frac{h_3^2}{2}\,\alpha_3 - \beta_3 - h_3\gamma_3 = 0$$

3)

$$\tau_1\ (y_1 = h_1) = \tau_2\ (y_2 = -h_2) \rightarrow$$
$$-\frac{\alpha_1}{2} h_1^2 - \beta_1 - \gamma_1 h_1 = -\frac{\alpha_2}{2} h_2^2 - \beta_2 + \gamma_2 h_2 \rightarrow$$
$$-\frac{h_1^2}{2} \alpha_1 + \frac{h_2^2}{2} \alpha_2 - \beta_1 + \beta_2 - h_1\gamma_1 - h_2\gamma_2 = 0$$

4)

$$\tau_2\ (y_2 = h_2) = \tau_3\ (y_3 = -h_3) \rightarrow$$
$$-\frac{\alpha_2}{2} h_2^2 - \beta_2 - \gamma_2 h_2 = -\frac{\alpha_3}{2} h_3^2 - \beta_3 + \gamma_3 h_3 \rightarrow$$
$$-\frac{h_2^2}{2} \alpha_2 + \frac{h_3^2}{2} \alpha_3 - \beta_2 + \beta_3 - h_2\gamma_2 - h_3\gamma_3 = 0$$

5)

$$-\int_{-h_1}^{h_1} \tau_1 dy_1 - \int_{-h_2}^{h_2} \tau_2 dy_2 - \int_{-h_3}^{h_3} \tau_3 dy_3 \rightarrow$$
$$\frac{\alpha_1}{2} I_1 + \beta_1 d_1 + \frac{\alpha_2}{2} I_2 + \beta_2 d_2 + \frac{\alpha_3}{2} I_3 + \beta_3 d_3 = P \rightarrow$$
$$\frac{I_1}{2} \alpha_1 + \frac{I_2}{2} \alpha_2 + \frac{I_3}{2} \alpha_3 + \beta_1 d_1 + \beta_2 d_2 + \beta_3 d_3 = P$$

6)

$$\epsilon_{x_1}\ (y_1 = h_1) = \epsilon_{x_2}\ (y_2 = -h_2) \rightarrow$$
$$\frac{\alpha_1}{E_1} x h_1 + \frac{\gamma_1}{E_1} x = -\frac{\alpha_2}{E_2} x h_2 + \frac{\gamma_2}{E_2} x \rightarrow$$
$$\frac{h_1}{E_1}\alpha_1 + \frac{h_2}{E_2}\alpha_2 + \frac{1}{E_1}\gamma_1 - \frac{1}{E_1}\gamma_2 = 0 \rightarrow$$

7)

$$\epsilon_{x_2}\ (y_2 = h_2) = \epsilon_{x_3}\ (y_3 = -h_3) \rightarrow$$
$$\frac{\alpha_2}{E_2} x h_2 + \frac{\gamma_2}{E_2} x = -\frac{\alpha_3}{E_3} x h_3 + \frac{\gamma_3}{E_3} x \rightarrow$$
$$\frac{h_1}{E_1}\alpha_2 + \frac{h_2}{E_2}\alpha_3 + \frac{1}{E_1}\gamma_2 - \frac{1}{E_1}\gamma_3 = 0$$

8) $$dV/dx = -\frac{V}{E}(\frac{\alpha}{2}y^2 + \gamma y) - \frac{\alpha x^2}{2E} + \frac{\alpha L^2}{2E} - \frac{2\beta(1+v)}{E}$$

$$\frac{dV_1}{dx}(y_1 = h_1) = \frac{dV_2}{dx}(y_2 = -h_2) \rightarrow$$

$$-\frac{V_1}{E_1}(\frac{\alpha_1}{2}h_1^2 + \gamma_1 h_1) + \frac{\alpha_1 L^2}{2\ E_1} - \frac{2\beta_1(1+v_1)}{E_1} =$$

$$-\frac{V_2}{E_2}(\frac{\alpha_2}{2}h_2^2 + \gamma_2 h_2) + \frac{\alpha_2 L^2}{2\ E_2} - \frac{2\beta_2(1+v_2)}{E_2} \rightarrow$$

$$-\frac{(v_1 h_1^2 - L^2)}{2E_1}\alpha_1 + \frac{(v_2 h_2^2 - L^2)}{2E_2}\alpha_2 - \frac{2(1+v_1)}{E_1}\beta_1 + \frac{2(1+v_2)}{E_2}\beta_2$$

$$-\frac{v_1 h_1}{E_1}\gamma_1 - \frac{v_1 h_1}{E_1}\gamma_2 = 0$$

9) $$\frac{dV_2}{dx}(y_2 = h_2) = \frac{dV_3}{dx}(y_3 = -h_3) \rightarrow$$

$$-\frac{V_2}{E_2}(\frac{\alpha_2}{2}h_2^2 + \gamma_2 h_2) + \frac{\alpha_2 L^2}{2\ E_2} - \frac{2\beta_2(1+v_2)}{E_2} =$$

$$-\frac{V_3}{E_3}(\frac{\alpha_3}{2}h_3^2 + \gamma_3 h_3) + \frac{\alpha_3 L^2}{2\ E_3} - \frac{2\beta_3(1+v_3)}{E_3} \rightarrow$$

$$-\frac{(v_2 h_2^2 - L^2)}{2E_2}\alpha_2 + \frac{(v_3 h_3^2 - L^2)}{2E_3}\alpha_3 - \frac{2(1+v_2)}{E_2}\beta_2 + \frac{2(1+v_3)}{E_3}\beta_3$$

$$-\frac{v_2 h_2}{E_2}\gamma_2 - \frac{v_3 h_3}{E_3}\gamma_3 = 0$$

Uit voorgaande 9 vergelijkingen kunnen m.b.v. matrix-berekening de 9 onbekenden berekend worden. De rekenmatrix voor de 3-lagenbalk is gegeven in bijlage A.

3. COMPUTERPROGRAMMA MULTIC

Voorgaande berekeningsmethode is basis voor het computerprogramma "MultiC", geschreven in Fortran 77.
Het programma bestaat uit een hoofdprogramma en 5 subroutines. Kort zal de functie van de genoemde onderdelen besproken worden.

* hoofdprogramma MULTIC

  Hierin is de invoer van de gegevens opgenomen. De elementen van de grondmatrix uit bijlage A. worden berekend en dienen als invoer voor de subroutine GAUSS. Met de berekende constanten $\alpha_{1..N}$, $\beta_{1..N}$ en $\gamma_{1..N}$ worden de spanningen en rekken in de verschillende lagen berekend.

* subroutine GAUSS

  Met behulp van de veegmethode van Gauss wordt de grondmatrix uit Multic omgevormd tot een driehoekige matrix

* subroutine MATRIX

  Uit de driehoekige matrix uit GAUSS worden de gewenste constanten α,β en γ berekend.

* subroutine MOMENT

  Het berekende spanningsbeeld wordt gecontroleerd door vergelijking van het inwendige moment M_i in de beschouwde doorsnede met het uitwendige moment M_u (= $P * L_x$)

* subroutine MULTEK

  Een tekeninvoer-file ("Multek.draw") voor het tekenen van het het rekverloop over de hoogte van de balk wordt gevuld.

* subroutine TAU

Berekend schuifspanningen in de balklagen en vult de tekeninvoer-file ("Tautek.draw") voor het tekenen van het schuifspanningsverloop over de hoogte van de beschouwde doorsnede.

4. VERGELIJKING MULTIC-SMITH.

Met het meerlagenmodel "Multic" moet het uiteraard mogelijk zijn om een 2-lagen constructie conform model "Smith" door te rekenen. Echter een directe vergelijking tussen Smith en Multic wordt bemoeilijkt door een verschil in werking.
Bij het model Multic is uitgegaan van een schuifvaste verbinding tussen alle lagen. Bij het Smith-model bevindt zich tussen de 2 lagen een schuiflaag met een variabele schuifstijfheidscoefficient. De werking hiervan is in het model gebaseerd op een relatie tussen een verschuiving van de 2 lagen t.o.v. elkaar en de schuifspanning in het contactvlak. De schuifstijfheidscoefficient Ki is gedefinieerd als het quotient van de schuifspanning en de verplaatsing:

$$Ki = \tau \,/\, s$$

In het model Multic is echter gesteld dat de verplaatsing van 2 lagen t.o.v. elkaar in het contactvlak 0 is. De rekken aan weerszijden van het contactvlak zullen dus gelijk aan elkaar zijn. Samengevat zien we voor de randvoorwaarden (6) en (7) uit hfdst. 2.2 de volgende verschillen:

Smith : $\tau_{n,xy} = Ki * (u_n - u_{n-1})$



Figuur 6. Vergelijking MULTIC-SMITH.

Multic : $\epsilon_{x,n} = \epsilon_{x,n-1}$

De vergelijking is te maken door bij het model Multic een dunne extra constructielaag in te voeren met een lage elasticiteitsmodulus. Bij bepaalde E/d verhoudingen van de extra laag zal een zelfde spanningsbeeld in een doorsnede verkregen worden als met gebruik van een schuifstijfheidscoefficient.
Hiervoor is een hulpprogramma TesTT (bijlage C.) geschreven voor een 2-laags constructie conform Multic, echter wel met een schuifstijfheidscoefficient Ki. Rekenresultaten zijn eerst vergeleken met die uit de publicatie [6].
Vervolgens is met TesTT en Multic het verband bepaald tussen Ki en de E/d verhouding wwarbij een zelfde spanningsverdeling optreedt. De dikte van de schuiflaag moet niet te groot gekozen worden aangezien een verschil optreedt in inwendige krachtsverdeling in een doorsnede.

4.1. Invoerdata vergelijking TesTT-Smith.

Bij de vergelijking tussen TesTT en Smith is uitgegaan van de proefgegevens uit de IABSE-publicatie [6],pag.11. Aangezien de ingevoerde belasting niet vermeld is, is deze voor een rekenvoorbeeld bepaald en daarna constant gehouden.

De ingevoerde data zijn:

Dikte staalplaat :	12 mm
Dikte asfalt :	40 mm
Elasticiteitsmod.staal:	$201,4*10^3$ N/mm^2
Contractiecoeff.staal :	0.3
" asfalt:	0.4
Belasting :	18.3 N/mm'
Lengte balk :	150 mm

S_{asfalt} [N/mm²]	Ki [N/mm³]
920	1.2
	4.0
	12.0
6400	10.0
	32.0
	100.0
17000	60.0
	200.0
	600.0

Uit vergelijking van de resultaten (figuur 7.) blijkt, dat de rekverdelingen berekend met het hulpprogramma TesTT goed overeenkomen met die van Smith [6].









Figuur 7-1. Vergelijking TesTT-Smith.









Figuur 7-2. Vergelijking TesTT-Smith.









Figuur 7-3. Vergelijking TesTT-Smith.

4.2. Vergelijking TesTT-Multic.

De vergelijking is uitgevoerd met de zelfde balk-geometrie als in voorgaande vergelijking. Er zijn 9 berekeningen uitgevoerd met 3 verschillende laagdikten en 3 verschillende E-moduli. De vergeleken balkopbouw uit de 2 modellen zijn weergegeven in fig. 8.



Figuur 8. Balkopbouw TesTT en Multic

Invoergegevens vergelijkingsberekening:

Laagdikte [mm]	A	B	C
asfalt	39.5	39.0	38.0
tussenlaag	0.5	1.0	2.0
staal	12.0	12.0	12.0

S-modulus asfalt	:	920,6400,17000 [N/mm²]
E-hechtlaag	:	te bepalen
E-staal	:	$201.4*10^3$ [N/mm²]
Belasting	:	18.3 N
Lengte balk	:	150 mm

Na berekening met de 2 programma's vinden we het volgende verband tussen Ki en E/d-hechtlaag:

$$Ki = 10^{(1.05\log E - 1.66\log D - 0.5)}$$

De 3 doorgerekende gevallen zijn weergegeven in fig. 9.



Figuur 9. Verband Ki en E/d-hechtlaag.

Het blijkt dus dat de werking van een schuiflaag of een dunne extra laag de zelde spanningsbeelden geven bij de juiste keuze. Het voordeel van Multic is echter dat er met materiaal- en diktegegevens gewerkt wordt en niet met een fictieve constante.

5. INVLOED LENGTE-INVOER OP BEREKENINGSRESULTATEN.

Aangezien bij de gevolgde rekenmethode vlakke loodrechte doorsneden bij belasting niet vlak blijven zal de lengte waarover deze vervorming zich kan ontwikkelen van invloed zijn op het spanningsverloop in de doorsnede.

In fig.10 is voor een rekenvoorbeeld het verloop van het verschil in rek boven en onder in laag 1 van een samengestelde balk weergegeven als functie van de ingevoerde balklengte L.





Figuur 10. Lengte-invloed op rekkenverloop.

Hier zien we dat bij een gelijke E-modulus van de balklagen de ingevoerde lengte geen invloed heeft op het spannings- cq. rekbeeld (hier: schuifvast verband tussen laag 1 en 3). Dit geldt ook bij juist zeer grote verschillen tussen de E-moduli (hier: schuiflos verband tussen laag 1 en 3). In alle andere situaties is de ingevoerde lengte L duidelijk van invloed op het spanningsverloop.
Bij het praktisch toepassen van het programma voor rek- en spanningsberekeningen is het dan ook van groot belang de juiste L-waarde te bepalen.

Voorbeeld:
Bij het controleren van gemeten rekken is het van belang de juiste rekken te bepalen in proefstukken. Hierbij beschouwen we 2 verschillende proefopstellingen, nl. A en B.(fig.11a-11b). Bij proefstuk A kan voor de ingevoerde lengte 150 mm worden genomen, voor proefstuk B zal deze waarde iets groter moeten zijn b.v. 180 mm. In beide gevallen moet de ingevoerde rekenlengte voor de beschouwde doorsnede gelijk blijven, nl. 150 mm, zodat het buigend moment in beide situaties gelijk is.



Figuur 11. Vergelijk 4-punts- en 3-puntsbuigproef.

6. WERKING/MOGELIJKHEDEN.

6.1. Algemeen.

Het is vooralsnog moeilijk te controleren of de berekende spanningen bij balken met sterk varierende laagstijfheden goed zijn. Hiervoor zouden proeven gedaan moeten worden met materialen met nauwkeuring bepaalde E-moduli. Zeker bij bitumineuze materialen zal dit een probleem zijn. Vergelijken we echter de vergelijking met de proefresultaten uit [6] met de berekende rekwaarden dan zien we toch een vrij grote nauwkeurigheid. We moeten echter zeker rekening houden met de lengte-invloed zoals vermeld in hfdst 5.0
Voor onderlinge vergelijkingen zal het programma zeker bruikbaar zijn.

6.2. Rekenvoorbeelden.

Met behulp van het programma is het mogelijk spanningsverdelingen in profielen te berekenen. De werking van het programma is gebaseerd op de onderlinge verhouding van laagstijfheden en werkt met een balkbreedte van 1. Door echter laagbreedten aan E-moduli te koppelen kunnen balken met varierende breedten worden doorgerekend. Bij invoer van de E-moduli worden deze vermeningvuldigd met de plaatselijke profielbreedte. Spanningsuitkomsten moeten vervolgens door de breedte worden gedeeld.

Voorbeeld: Maximale optredende spanning in HE 100 B profiel

profielgegevens: $b = 100$ mm

$a = 6$ mm

$e = 10$ mm

$W_x = 90.000$ mm^3

HE 100 B

Invoer programma : zie Bijlage D.

Max. spanning $\sigma_x = M/W = 10.000/90.000 = 0.111$

Computerberekening: σ_x = tenstop(1) $= 11.61/b$

$= 0.116$

Voorbeeld: Vergelijking rekken in slijtlaagconstructie met ZOAB t.o.v. een traditionle opbouw.

NOG VERDER BESCHRIJVEN !!!!! FIGUREN: ZIE BIJLAGE E.

7. REFERENTIES

[1] De Back, J.;Span, H.J.Th.;Van Schooten, J.,"Optimalisatie van de economische bruikbaarheid van dikke slijt- en isolatielagen op stalen brugdekken (criteria en factoren)", Aanvraag ZWO Projectfinanciering, Faculteit der Civiele Techniek, Technische Universiteit Delft, 1983.

[2] Kolstein, M.H.;Fontijn, H.F.N.;Kok, J.C.,1988, "Vermoeiingsgedrag van gemodificeerd gietasfalt op stalen orthotrope brugdekken", Stevinrapport 25.6.89.2., Delft University of Technology, Stevinlaboratory-Steelstructures, Delft.

[3] Kolstein, M.H.;Fontijn, H.F.N.;Kok, J.C.,1988,"Theoretical models and thickness effect of asphalt surfacing on steel highway bridges", Stevinreport 25.6.89.3, Delft University of Technology, Stevinlaboratory-Steelstructures, Delft.

[4] Kolstein, M.H.;Dijkink, J.H.;1989,"Behaviour of modified bituminous surfacings on orthotropic steel bridge decks". Submitted for publication on 4th Eurobitume, Madrid.

[5] Timoshenko,S.P.;Goodier,J.N.;1970;"Theory of Elasticity

[6] Smith,J.W.;Cullimore,M.S.G.;Flett,I.P.;1983;"Flexure of Steel Bridge Deck Plate with Asphalt Surfacing",IABSE-preriodica 1/1983

BIJLAGE A. Reken-matrix "Multic" (3-laags-constructie).

	A	B	C	D	E	F	G	H	I
1	$\frac{-H(1)^2}{2}$	0	0	-1	0	0	$H(1)$	0	0
2	0	0	$\frac{-H(3)^2}{2}$	0	0	-1	0	0	$-H(3)$
3	$\frac{-H(1)^2}{2}$	$\frac{H(2)^2}{2}$	0	-1	1	0	$-H(1)$	$-H(2)$	0
4	0	$\frac{-H(2)^2}{2}$	$\frac{H(3)^2}{2}$	0	-1	1	0	$-H(2)$	$-H(3)$
5	$\frac{I(1)}{2}$	$\frac{I(2)}{2}$	$\frac{I(3)}{2}$	$D(1)$	$D(2)$	$D(3)$	0	0	0
6	$\frac{H(1)}{E(1)}$	$\frac{H(2)}{E(2)}$	0	0	0	0	$\frac{1}{E(1)}$	$\frac{-1}{E(2)}$	0
7	0	$\frac{H(2)}{E(2)}$	$\frac{H(3)}{E(3)}$	0	0	0	0	$\frac{1}{E(2)}$	$\frac{-1}{E(3)}$
8	$\frac{-v(1)*H(1)^2+L^2}{2*E(1)}$	$\frac{v(2)*H(2)^2-L^2}{2*E(2)}$	0	$\frac{2+2*v(1)}{E(1)}$	$\frac{2+2*v(2)}{E(2)}$	0	$\frac{-v(1)*H(1)}{E(1)}$	$\frac{-v(2)*H(2)}{E(2)}$	0
9	0	$\frac{-v(2)*H(2)^2+L^2}{2*E(2)}$	$\frac{v(3)*H(3)^2-L^2}{2*E(3)}$	0	$\frac{2+2*v(2)}{E(2)}$	$\frac{2+2*v(3)}{E(3)}$	0	$\frac{-v(2)*H(2)}{E(2)}$	$\frac{-v(3)*H(3)}{E(3)}$

Reken-matrix " MULTIC " (3-laags-constr.)

BIJLAGE B. Programma MULTIC

```
*---------------------------------------------------------------C
*   MULTIC.CMD     Dirk Wiglema                                C
*---------------------------------------------------------------C
FTN7X MULTIC.INC,MULTIC.LST,MULTIC.REL
LINK MULTIC.LOD
```

```
*-------------------------------------------------------C
*     MULTIC.INC       Dirk Wiglema                    C
*-------------------------------------------------------C
$INCLUDE multic.FTN
$INCLUDE GAUSS.FTN
$INCLUDE MATRIX.FTN
$INCLUDE MOMENT.FTN
$INCLUDE TAU.FTN
$INCLUDE MULTEK.FTN
```

```
*------------------------------------------------------------C
*    MULTIC.LOD      Dirk Wiglema    23-6-88   <890424.1357>
*------------------------------------------------------------C
LL,MULTIC.MAP
RE MULTIC.REL             (hoofdprogramma+subroutines)
EN
```

```
$FILES 0,4
*****************************************************************
      PROGRAM MULTIC (), Dirk Wiglema  <890424.1433>
*****************************************************************
      IMPLICIT REAL (A-Z)
      REAL V(60,60),W(60),FORCE(60),H(20),E(20),D(20),VV(20)
      REAL TENSTOP(20),TENSBOT(20),STRTOP(20),STRBOT(20),II(20)
      REAL XXX(15)
      INTEGER k,kk,kkk,n,LUO,LUO1,m,NR,S
      CHARACTER*40 NAAM1                 ! data-file invoergegevens
      CHARACTER*40 NAAM2                 ! file-naam uitvoerfile
C
C****************************************************************
C
C      lijst van gebruikte variabelen:
C
C      V(i,j)    :  matrix-invoerwaarden
C      W(i)      :  hulpwaarden spanningsberekening
C      FORCE(i):    beeld-matrix van V(i,j)
C      H(i)      :  halve dikte laag i
C      D(i)      :  dikte laag i
C      II(i)     :  traagheidsmoment laag i
C      E(i)      :  elasticiteitsmodulus laag i
C      VV(i)     :  contractiecoefficient laag i
C      TENSTOP(i)   :  spanning bovenzijde laag i
C      TENSBOT(i)   :  spanning onderzijde laag i
C      STRTOP(i)    :  rek bovenzijde laag i
C      STRBOT(i)    :  rek onderzijde laag i
C      n            :  aantal lagen
C      m            :  matrix afmeting ( 3 x n )
C      k,kk,kkk     :  hulpvariabelen
C      luo1         :  lu-nummer voor printuitvoer
C      NR           :  berekeningsnummer
C      NAAM1        :  naam uitvoerfile van de berekeningsgegevens
C      NAAM2        :  naam uitvoerfile (bv. tekenfile)
C      S            :  var. control-loop
C
C****************************************************************
C
      WRITE(1,*) ' ****************************************************'
      WRITE(1,'(6X,"PROGRAMMA    ***  MULTIC   ***")')
      WRITE(1,*) '  programma voor het berekenen van spanningen en '
      WRITE(1,*) '  rekken in een meerlagige balk                  '
      WRITE(1,*) ' ****************************************************'
      WRITE(1,*) '   '
      WRITE(1,'(6X,"                     Berekeningsvolgnummer ? _")')
      READ(1,'(I6)')NR
C
      LUO1 = 1
C
1     WRITE(1,'(6X,"           Geprinte uitvoer gewenst (ja=1) ? _")')
      S=0
      READ(1,'(I2)')S
      IF (S.EQ.1) LUO1=6
C
      WRITE(1,'(6X,"           Uitvoergegevens naar tekenfile :  _")')
      READ(1,'(A40)')NAAM2
      WRITE(1,*) '                          tekenfile : ',naam2, '   '
C
      OPEN(47,FILE=NAAM2)
```

```
      REWIND 47
C
      WRITE(1,*) '    '
      WRITE(1,*) '    '
      WRITE(1,*) '    '
      WRITE(1,*) ' Invoer berekeningsgegevens '
      WRITE(1,*) ' *********************************************'
      WRITE(1,*) '  '
333   WRITE(1,'(6X,"                        aantal lagen  ? _")')
      READ(1,'(I3)')n
C
      IF (N.GT.20) THEN
          WRITE(1,*) ' aantal opgegeven lagen te groot (<20) '
          GOTO 333
      END IF
C
      DO 05 i =1,n
112     S=0
        WRITE(1,*) '   '
        WRITE(1,'(6X,"  laag ",I3)')i
        WRITE(1,'(20X,"           dikte : _")')
        READ(1,'(F10.2)')D(i)
        WRITE(1,'(20X,"           e-mod : _")')
        READ(1,'(F10.2)')E(i)
        WRITE(1,'(20X,"     contr.coeff : _")')
        READ(1,'(F10.2)')VV(i)
        WRITE(1,'(6X,"                        juist (nee=1) ? _")')
        READ(1,'(I2)')S
        IF(S.EQ.1) GOTO 112
05    CONTINUE
C
113   S=0
      WRITE(1,*) '   '
      WRITE(1,'(6X,"         lengte balk      [mm] :_")')
      READ(1,'(F10.2)')L
      WRITE(1,'(6X,"         rekenlengte balk [mm] :_")')
      READ(1,'(F10.2)')LX
      WRITE(1,'(6X "         belasting  [N]        :_")')
      READ(1,'(F10.2)')P
      WRITE(1,'(6X,"                        juist (nee=1) ? _")')
      READ(1,'(I2)')S
      IF(S.EQ.1) GOTO 113
C
      k = n+2
      kk = 2*n
      kkk = kk+1
      kkkk = n*3
      m = kkkk
C
121   DO 10 i = 1,n
        H(i) = D(i)/2
        II(i) = D(i)**3/12
10    CONTINUE
C
      DO 20 i = 1,kkkk
        W(i) = 0
        FORCE(i) = 0
20    CONTINUE
      FORCE(k) = P
C
```

```
      DO 30 i = 1,kkkk
        DO 40 j = 1,kkkk
          V(i,j) = 0
 40     CONTINUE
 30   CONTINUE
 C
      V(1,1)    =  -H(1)**2/2
      V(1,n+1)  =  -1.
      V(1,kk+1) =   H(1)
      V(2,n)    = -H(n)**2/2
      V(2,kk)   = -1.
      V(2,kkkk) = -H(n)
 C
      DO 60 I = 1,n-1
        V(i+2,i)        =  -H(i)**2/2
        V(i+2,i+1)      =   H(i+1)**2/2
        V(i+2,i+n)      =   -1.
        V(i+2,i+n+1)    =    1.
        V(i+2,kk+i)     =   -H(i)
        V(i+2,kk+i+1)   =   -H(i+1)
 60   CONTINUE
 C
      DO 70 i = 1,n
        V(k,i)   = II(i)/2
        V(k,i+n) = D(i)
 70   CONTINUE
 C
      DO 80 i = 1,n-1
        V(k+i,i)        =  H(i)/E(i)
        V(k+i,i+1)      =  H(i+1)/E(i+1)
        V(k+i,kk+i)     =  1/E(i)
        V(k+i,kk+i+1)   = -1/E(i+1)
 80   CONTINUE
 C
      DO 90 i = 1,n-1
        V(kkk+i,i)        = -(VV(i)*H(i)**2-L**2)/(2*E(i))
        V(kkk+i,i+1)      =  (VV(i+1)*H(i+1)**2-L**2)/(2*E(i+1))
        V(kkk+i,n+i)      = -(2.+2*VV(i))/E(i)
        V(kkk+i,n+i+1)    =  (2.+2*VV(i+1))/E(i+1)
        V(kkk+i,kk+i)     = -(VV(i)*H(i))/E(i)
        V(kkk+i,kk+i+1)   = -(VV(i+1)*H(i+1))/E(i+1)
 90   CONTINUE
 C
      CALL GAUSS(M,V,FORCE,LUO)
      CALL MATRIX(V,W,FORCE,M)
 C
      DO 100 i = 1,n
        TENSTOP(i) =  (-W(i)*H(i) + W(i+kk))*Lx
        TENSBOT(i) =   (W(i)*H(i) + W(i+kk))*Lx
        STRTOP(i)  =  1000000. * TENSTOP(i)/E(i)
        STRBOT(i)  =  1000000. * TENSBOT(i)/E(i)
 100  CONTINUE
 C
      WRITE(1,'(6X,"Invoergegevens wegschrijven naar file : _")')
      READ(1,'(A40)') NAAM1
      WRITE(1,*) '                    data-file = ',naam1,'    '
 C

      WRITE(LUO1,*) '     '
      WRITE(LUO1,*) ' **** M   U   L   T   I   C ****'
```

```
      WRITE(LUO1,*) ' Invoer-data balkberekening ',NR,'  '
      WRITE(LUO1,*) '-----------------------------------------------'
      WRITE(LUO1,'(6X,"lengte balk          = ",F10.0," mm" )')L
      WRITE(LUO1,'(6X,"rekenlengte balk     = ",F10.0," mm")')Lx
      WRITE(LUO1,'(6X,"belasting            = ",F10.0," N")')P
      WRITE(LUO1,*) '  '
      DO 110 i = 1,n
        WRITE(LUO1,'(6X,"LAAG  ",I3)')I
        WRITE(LUO1,'(15X,"  laagdikte    = ",F10.0," mm")')D(i)
        WRITE(LUO1,'(15X,"  E-modulus    = ",F10.0," N/mm2")')E(i)
        WRITE(LUO1,'(15X,"  contr.coeff = ",F10.3)')UU(i)
 110  CONTINUE
 C
      WRITE(LUO1,*) ' ****** UITVOER BEREKENING "MULTIC" ****** '
      WRITE(LUO1,*) '  '
      WRITE(LUO1,*) '   Berekeningsnummer :  ',NR, '  '
      WRITE(LUO1,*) '  '
      WRITE(LUO1,*) '   Berekende spanningen en rekken per laag '
      WRITE(LUO1,*) '  '
      WRITE(LUO1,'(4X,"laag     tenstop   tensbot        strtop    strbo
     >t ")')
      WRITE(LUO1,'(14X,"-------------------------------------------")')
 C
      DO 130 i = 1,n
      WRITE(LUO1,'(6X,I3,")  ",2F10.3,5X,2F10.3)')I,TENSTOP(i),
     >                         TENSBOT(i),STRTOP(i),STRBOT(i)
 130  CONTINUE
      WRITE(LUO1,*) '  '
      WRITE(LUO1,*) '  '
 C
      CALL MOMENT(TENSTOP,TENSBOT,D,Lx,P,LUO1,N)
 C    CALL MULTEK(D,STRTOP,STRBOT,N)
 C    CALL TAU(W,D,E,N)
 C
      CLOSE(47)
 C
      WRITE(1,'(6X,"           nieuwe berekening (ja=1) ? _")')
      READ(1,'(I2)')S
      IF(S.EQ.1) GOTO 1
 C
      WRITE(LUO,'(6X,"****   EINDE PROGRAMMA   ****")')
 C
      END
```

```
      SUBROUTINE GAUSS(M,TDC,FORCEC,LUO)
C     ******************************
C     GAUSS ELIMINATION TECHNIQUE APPLIED TO TDC TO DERIVE AN UPPER
C     TRIANGULATED MATRIX
C     ROW PIVOTING IS USED TO MINIMIZE THE LOSS OF ACCURACY AND TO
C     PREVENT THAT ANY DIAGONAL ELEMENT WILL BE 0
C
      REAL COUNT,MULT
      DIMENSION TDC(60,60),FORCEC(60)
      INTEGER LUO,P,M
      DO 10 I=1,M-1
         P=I
         DO 20 J=I,M-1
            IF (ABS(TDC(P,I)).LT.ABS(TDC(J+1,I))) THEN
               P=J+1
            END IF
 20      CONTINUE
         DO 30 II=I,M
            COUNT= TDC(P,II)
            TDC(P,II)=TDC(I,II)
            TDC(I,II)=COUNT
            IF (II.EQ.I) THEN
               COUNT = FORCEC(P)
               FORCEC(P)=FORCEC(I)
               FORCEC(I)=COUNT
            END IF
 30      CONTINUE
         DO 40 JJ=I,M-1
C
            IF (TDC(I,I).EQ.0D0) THEN
               WRITE(LUO,'(6X,"ERROR FOUND IN SUBR. GAUSS",/,6X,
     >                              "CHECK MATRIX")')
               STOP
            END IF
C
            MULT=TDC(JJ+1,I)/TDC(I,I)
            FORCEC(JJ+1)=FORCEC(JJ+1)-MULT*FORCEC(I)
            DO 50 II=I,M
               TDC(JJ+1,II)=TDC(JJ+1,II)-MULT*TDC(I,II)
 50         CONTINUE
 40      CONTINUE
 10   CONTINUE
      END
```

```
C*************************************************************************
      SUBROUTINE MATRIX(FUNC,SOL,FORCED,M),<890410.0946>
C*************************************************************************
C     SUBR. TO DERIVE A SOLUTION TO A TRIANGULAR MATRIX
C-------------------------------------------------------------------------
      REAL SOM
      REAL FUNC(60,60),SOL(60),FORCED(60)
      INTEGER M
C
      SOL(M)=FORCED(M)/FUNC(M,M)
C
      DO 10 I=M-1,1,-1
         SOM=0
         DO 20 K=M,I+1,-1
             SOM=SOM + FUNC(I,K)*SOL(K)
20       CONTINUE
         SOL(I)= (FORCED(I)-SOM)/FUNC(I,I)
10    CONTINUE
C
      END
```

```
C------------------------------------------------------------------------
      SUBROUTINE MOMENT(TT,TB,D,LX,P,LUO1,N),    23-06-88 <890418.1038
C------------------------------------------------------------------------
      IMPLICIT REAL (A-Z)
      REAL TT(20),TB(20),MOM(20,2),D(20)
      INTEGER LUO1,n
C
C***********************************************************************
C     variabelen:
C
C     TT           : TENSTOP  = spanning bovenzijde laag
C     TB           : TENSBOT  =     "     onder-  "    "
C     D            : laagdikte
C     LX           : lengte vrije uiteinde tot beschouwde drsn.
C     P            : belasting
C     luo1         : LU-nummer
C     N            : aantal lagen
C     MOM          : moment in drsn.
C***********************************************************************
C     controle:som van krachten over de hoogte van de
C              doorsnede is 0
      SOM = 0
      DO 301 I=1,N
          SOM = SOM + (TT(I)+TB(I))*D(I)/2
 301  CONTINUE
      WRITE(LUO1,*) ' ******* UITVOER SUBROUTINE "MOMENT" ****** '
      WRITE(LUO1,*) '  '
      WRITE(LUO1,'(6X,"som van horizontale krachten = : ",F10.5)')SOM
      IF (SOM.GT.-0.005.AND.SOM.LT.0.005) THEN
C
      DHULP=0
      MTOT=0
      DO 401 I=1,N
C
         HULP1=TT(I)-ABS(TT(I))
         HULP2=TB(I)-ABS(TB(I))
C
        IF (HULP1.EQ.0.AND.HULP2.EQ.0.OR.HULP1.NE.0.AND.HULP2.NE.0)
     >  THEN
            IF (ABS(TT(I)).GT.ABS(TB(I))) THEN
              MOM(I,1)= TB(I)*D(I)*(D(I)/2+DHULP)+(TT(I)
     >                 -TB(I))*D(I)*(D(I)/3+DHULP)/2
              MOM(I,2)=0
            END IF
C
            IF (ABS(TT(I)).LE.ABS(TB(I))) THEN
              MOM(I,1)= TT(I)*D(I)*(D(I)/2+DHULP)+(TB(I)
     >                 -TT(I))*D(I)*(D(I)/1.5+DHULP)/2
              MOM(I,2)=0
            END IF
      ELSE
          A=D(I)*TT(I)/ABS(TT(I)-TB(I))
          B=D(I)-A
          MOM(I,1)= TT(I)/2*A*(A/3+DHULP)
          MOM(I,2)= TB(I)/2*B*(B/1.5+A+DHULP)
      END IF
C
         DHULP=D(I)+DHULP
         MTOT=MTOT+MOM(I,1)+MOM(I,2)
 401  CONTINUE
```

```
C
              MCONTR=-P*Lx
C
        WRITE(LUO1,'(42X,"MTOT =      ",F12.5)')MTOT
        WRITE(LUO1,'(42X,"MCONTR. = ",F12.5,/)')MCONTR
        DELTAM=MTOT-MCONTR
        IF(ABS(DELTAM).LT.1.) GOTO 222
C
      END IF
 111    WRITE(LUO1,'(6X,"Error in tension-calculation")!)
C
 222  WRITE(LUO1,*) '    '
      WRITE(LUO1,*) ' Beeindiging subroutine "MOMENT" '
      WRITE(LUO1,*) '    '
      END
```

```
$FILES 0,2
C********************************************************************
C********************************************************************
      SUBROUTINE TAU(WW,DD,EE,N)
C********************************************************************
      REAL TAUW,Y,H
      REAL WW(60),DD(20),EE(20)
      INTEGER N,k,kk
C
      OPEN (48,FILE='TAUTEK.DRAW')
      REWIND 48
C
      k=n
      kk=2*n
      H=0.
      DO i = 1,n
        H=H + DD(i)
      END DO
C
      DO 10 I=1,n
         Y=-.5*DD(i)
         WRITE(1,'(6X," LAAG (",I2,")")')I
         DO 20 J=1,11

C
         TAUW = (WW(i)*Y**2/2+WW(i+k)+WW(i+kk)*Y)
         WRITE(48,'(T11,F10.6,T21,F10.6)')H,TAUW
         Y=Y+.1*DD(i)
         H=H-.1*DD(i)
C
 20   CONTINUE
         H=H+.1*DD(i)
 10   CONTINUE
      CLOSE (48)
      WRITE(1,*) '   '
      WRITE(1,*) ' invoerfile TAUTEK gevuld ! '
      END
```

```
C*********************************************************************
      SUBROUTINE MULTEK(D,STRT,STRB,N), Wiglema 310888 <890418.1131>
C*********************************************************************
C       Subroutine voor vullen van invoer-file tbv. programma
C       GRAPH. Tekent rek-figuren in balkdoorsnede.
C
      IMPLICIT REAL(A-Z)
      REAL D(20),STRT(20),STRB(20),A(21)
      INTEGER PPP,n
C
      OPEN(49,FILE='multekin')
      REWIND 49
      WRITE(1,*) ' Invoer-file voor GRAPH :  MULTEKIN '
C
      DO i = 1,n
        A(i) = 0
        DO j = i,n
          A(i) = A(i) + D(j)
        END DO
      END DO
      A(1+n) = 0
C
      DO 10 I=1,n
         WRITE(49,'(T21,F6.3,T31,F6.1)')STRT(i),A(i)
         WRITE(49,'(T21,F6.3,T31,F6.1)')STRB(i),A(i+1)
 10   CONTINUE
C
      CLOSE(49)
C
      WRITE(1,*) '                    '
      WRITE(1,'("     subr. multek beeindigd    |")')
      WRITE(1,'("     inputfile MULTEKIN gevuld   |")')
      WRITE(1,*) '                '
      END
```

BIJLAGE C. Hulpprogramma TesTT

Rekenmatrix uit TesTT

```
V(1,1) = -H1**2/2
V(1,2)=0
V(1,3) = -1.
V(1,4)=0
V(1,5) = H1
V(1,6)=0
V(2,1)=0
V(2,2) = -H2**2/2
V(2,3)=0
V(2,4) = -1.
V(2,5)=0
V(2,6) = -H2
V(3,1) = -H1**2/2
V(3,2) =  H2**2/2
V(3,3) = -1.
V(3,4) = 1.
V(3,5) = -H1
V(3,6) = -H2
V(4,1) = I1/2
V(4,2) = I2/2
V(4,3) = D1
V(4,4) = D2
V(4,5)=0
V(4,6)=0
V(5,1) = .5*H1**2+KI*((2.+V1)*H1**3/(6*E1)+(L**2*H1)/(2*E1))
V(5,2) =          KI*((2.+V2)*H2**3/(6*E2)+(L**2*H2)/(2*E2))
V(5,3)=1
V(5,4)=0
V(5,5) =       H1+KI*((2.+V1)*H1**2/(2*E1)+L**2/(2*E1))
V(5,6) =         -KI*((2.+V2)*H2**2/(2*E2)+L**2/(2*E2))
V(6,1) = -(V1*H1**2-L**2)/(2*E1)
V(6,2) =  (V2*H2**2-L**2)/(2*E2)
V(6,3) = -(2.0+2*V1)/E1
V(6,4) =  (2.0+2*V2)/E2
V(6,5) = -V1*H1/E1
V(6,6) = -V2*H2/E2
```

BIJLAGE D. Rekenvoorbeeld HE 100 B

```
**** M   U   L   T   I   C ****
Invoer-data balkberekening  1
-----------------------------------------------
    lengte balk          =      1000. mm
    rekenlengte balk     =      1000. mm
    belasting            =        10. N

    LAAG    1
                laagdikte    =        10.  mm
                E-modulus    =       100.  N/mm2
                contr.coeff  =       .300
    LAAG    2
                laagdikte    =        80.  mm
                E-modulus    =         6.  N/mm2
                contr.coeff  =       .300
    LAAG    3
                laagdikte    =        10.  mm
                E-modulus    =       100.  N/mm2
                contr.coeff  =       .300
****** UITVOER BEREKENING "MULTIC" ******

  Berekeningsnummer :    1

  Berekende spanningen en rekken per laag

  laag      tenstop   tensbot          strtop     strbot
            ---------------------------------------------
     1)      11.611      9.212      116109.160 92122.906
     2)        .553      -.553       92122.687-92123.359
     3)      -9.212    -11.611      -92122.781-116109.05


******* UITVOER SUBROUTINE "MOMENT" ******

    som van horizontale krachten = :      -.00003
                                          MTOT =     -10000.00400
                                          MCONTR. =  -10000.00000


Beeindiging subroutine "MOMENT"
```

BIJLAGE E. Rekenvoorbeeld ZOAB-slijtlaag

```
**** M   U   L   T   I   C ****
Invoer-data balkberekening  2
--------------------------------------------
    lengte balk          =       170.  mm
    rekenlengte balk     =       130.  mm
    belasting            =        10.  N

    LAAG     1
              laagdikte      =        20.  mm
              E-modulus      =     10000.  N/mm2
              contr.coeff    =       .400
    LAAG     2
              laagdikte      =        25.  mm
              E-modulus      =     10000.  N/mm2
              contr.coeff    =       .300
    LAAG     3
              laagdikte      =         6.  mm
              E-modulus      =      1000.  N/mm2
              contr.coeff    =       .400
    LAAG     4
              laagdikte      =         1.  mm
              E-modulus      =        10.  N/mm2
              contr.coeff    =       .400
    LAAG     5
              laagdikte      =        12.  mm
              E-modulus      =    210000.  N/mm2
              contr.coeff    =       .300
****** UITVOER BEREKENING "MULTIC" ******

  Berekeningsnummer :    2

  Berekende spanningen en rekken per laag

  laag      tenstop    tensbot        strtop     strbot
           ------------------------------------------------
     1)       2.067       .482        206.716     48.162
     2)        .482     -1.473         48.163   -147.298
     3)       -.147      -.182       -147.298   -182.039
     4)       -.002       .000       -182.038     44.344
     5)       9.312    -11.330         44.344    -53.953


******* UITVOER SUBROUTINE "MOMENT" ******

    som van horizontale krachten = :      .00000
                                      MTOT =       -1299.90770
                                      MCONTR. =    -1300.00000


Beeindiging subroutine "MOMENT"
```

berekeningsnr. 001

Vergelijking rekken in slijtlaagconstructie met ZOAB t.o.v. een trad. opbouw

ZOAB OPBOUW

gegevens: laag	1	ZOAB	d = 35 mm
	2	hechtl.	d = 5 mm
	3	gietasf.	d = 25 mm
	4	hechtl.	d = 5 mm
	5	staalpl.	d = 12 mm

TRADITIONELE OPBOUW

gegevens: laag	1	gietasf.	d = 22 mm
	2	"	d = 20 mm
	3	mastiek	d = 7 mm
	4	kleef	d = 1 mm
	5	staalpl.	d = 10 mm

Berekening uitgevoerd met " **MULTIC** ":

gegevens: F = 10 N   L = 170 mm
M = 1300 Nmm   Lx = 130 mm



Stevin : 25.6.89.13/A2

TU-Delft

berekeningsnr. 004 1

Vergelijking rekken in slijtlaagconstructie met ZOAB t.o.v. een trad. opbouw

ZOAB OPBOUW

gegevens: laag 1	ZOAB	d = 35 mm	
2	hechtl.	d = 5 mm	
3	gietasf.	d = 25 mm	
4	hechtl.	d = 5 mm	
5	staalpl.	d = 12 mm	

TRADITIONELE OPBOUW

gegevens: laag 1	gietasf.	d = 22 mm
2	"	d = 20 mm
3	mastiek	d = 7 mm
4	kleef	d = 1 mm
5	staalpl.	d = 10 mm

Berekening uitgevoerd met " **MULTIC** ":

gegevens: F = 10 N L = 170 mm
M = 1300 Nmm Lx = 130 mm



berekeningsnr. 003

Vergelijking rekken in slijtlaagconstructie met ZOAB t.o.v. een trad. opbouw

ZOAB OPBOUW

gegevens:	laag		
	1	ZOAB	d = 35 mm
	2	hechtl.	d = 5 mm
	3	gietasf.	d = 25 mm
	4	hechtl.	d = 5 mm
	5	staalpl.	d = 12 mm

TRADITIONELE OPBOUW

gegevens:	laag		
	1	gietasf.	d = 22 mm
	2	"	d = 20 mm
	3	mastiek	d = 7 mm
	4	kleef	d = 1 mm
	5	staalpl.	d = 10 mm

Berekening uitgevoerd met " **MULTIC** ":

gegevens: F = 10 N L = 170 mm
M = 1300 Nmm Lx = 130 mm



berekeningsnr. 004

Vergelijking rekken in slijtlaagconstructie met ZOAB t.o.v. een trad. opbouw

ZOAB OPBOUW

gegevens:	laag		
	1	ZOAB	d = 35 mm
	2	hechtl.	d = 5 mm
	3	gietasf.	d = 25 mm
	4	hechtl.	d = 5 mm
	5	staalpl.	d = 12 mm

TRADITIONELE OPBOUW

gegevens:	laag		
	1	gietasf.	d = 22 mm
	2	"	d = 20 mm
	3	mastiek	d = 7 mm
	4	kleef	d = 1 mm
	5	staalpl.	d = 10 mm

Berekening uitgevoerd met " **MULTIC** ":

gegevens: F = 10 N, L = 170 mm
M = 1300 Nmm, Lx = 130 mm